
log using "D:\Jason\workshop\Stata workship\Stata presentation\stata presentation5_part1.log", replace

*************************************************************************************************
* This command files has three aims
* 1. Demonstrate how to use -table-, -dtable-, and -etable- to quickly generate different tables
* 2. Demonstrate how to use -table- and -collect- commands to generate and customize tables
* 3. Shows how to use -graph- commands to generate and customize graphs
**************************************************************************************************

**********************************************************************************
* Aim #1: Use -table-, -dtable-, and -etable- to quickly generate different tables
**********************************************************************************

webuse nhanes2l, clear
describe  age sex diabetes race heartatk
sum heartatk age sex diabetes race

  ************************************
  * Frequencies and Percentages
  ************************************

    *************************
    * One-way tables
    ***************************
tab   race
table race
table race, statistic(frequency) statistic(percent)
collect label levels result frequency "N" percent "%", modify
collect preview

    *******************************
    * Two-Way Tables
    *******************************
table race sex, statistic(percent, across(sex))   
table sex  race,  statistic(percent, across(sex)) 

    ****************************
    * Multi-way Tables
    ****************************  
 
table (race sex) (diabete), statistic(percent, across(diabete))

    ****************************
    * Summary Statistics
    ***************************

table race sex, statistic(mean age) nformat(%5.2f)    

collect title "Mean age by race and sex"
collect preview

table race sex, statistic(mean age) statistic(sd age) ///
   nformat(%5.2f) sformat((%s) sd)
collect style header result, level(hide)
collect preview



  ***************************
  * Descriptive Statistics
  ***************************

dtable age i.race i.diabetes, by(sex, test)

gen N = 1
table (var) (sex) ,   ///
   stat(count N)                         ///  /*sample*/
   stat(mean age) stat(sd age)           /// /*continuous*/
   stat(fvpercent race diabetes)                  /// /*factor*/
   nformat(%5.2f mean sd) nformat(%5.1f fvpercent) ///
   sformat((%s) sd) sformat(%s%% fvpercent)  style(table-1)

  ***************************
  * Tables of Estimates
  **************************

    ********************
    * Single Regression
    ********************

logistic diabetes sex age 
etable, showstars showstarsnote

    ********************
    * Tables of margins
    ********************

logistic diabetes i.sex age i.race
margins i.sex i.race

*Create a table with the average predicted probabilities and confidence intervals estimated by margins
etable, margins cstat(_r_b) cstat(_r_ci)

*Same as above, but use a comma as the delimiter between confidence interval limits and hide the number of observations
etable, margins cstat(_r_b) cstat(_r_ci, cidelimiter(,)) mstat(_hide)

    ************************
    * Compare two regression
    ***********************

logistic diabetes sex age 
estimates store model_1

logistic diabetes sex age i.race
estimates store model_2

etable, estimates(model_1 model_2) column(estimates) ///
     cstat(_r_b) cstat(_r_z, sformat((%s))) ///
     note(test statistic in parentheses) showstars showstarsnote

    *************************************
    *Regressions with Different Outcomes
    *************************************

logistic diabetes sex age i.race
estimates store diabetes

logistic heartatk sex age i.race
estimates store heartatk

etable, estimates(diabetes heart_attack) eqrecode(diabetes=both heartatk=both)  mstat(N) mstat(r2) showstars showstarsnote 



log close

*******************************************************************************************
* Aim #2. Demonstrate how to use -table- and -collect- commands to generate and customize tables
********************************************************************************************

log using "D:\Jason\workshop\Stata workship\Stata presentation\stata presentation5_part2.log", replace

***********************************************************************
* A more complex example of using results of three regressions to 
* create a customized tables
***********************************************************************

webuse nhanes2l, clear

collect clear
collect create MyModels

******************************************
* Get the statistics from the regressin
******************************************

  ************************
  * First Regression Model
  ************************

collect, name(MyModels): logistic heartatk c.age i.sex

collect _r_b _r_se,                    ///
        name(MyModels)                 ///
        tag(model[(1)])                ///
        : logistic heartatk c.age i.sex

collect layout (colname#result) (model[(1)]), name(MyModels)

collect dims
collect levelsof result
collect layout (result), name(MyModels)
collect levelsof colname
collect layout (colname) (result), name(MyModels)
collect layout (colname#result) (), name(MyModels)
collect levelsof model
collect layout (colname#result) (model[(1)]), name(MyModels)

estat ic
return list
matlist r(S)
display r(S)[1,"BIC"]

collect AIC=r(S)[1,"AIC"]    ///
        BIC=r(S)[1,"BIC"],   ///
        name(MyModels)       ///
        tag(model[(1)])      ///
        : estat ic



collect layout (colname#result result[AIC BIC]) (model[(1)]), name(MyModels)

  **********************
  * Add a second model
  **********************

collect _r_b _r_se,                    ///
        name(MyModels)                 ///
        tag(model[(2)])                ///
        : logistic heartatk c.age##i.sex

collect AIC=r(S)[1,"AIC"]              ///
        BIC=r(S)[1,"BIC"],             ///
        name(MyModels)                 ///
        tag(model[(2)])                ///
        : estat ic

collect layout (colname#result result[AIC BIC]) (model), name(MyModels)

collect preview

  *********************
  * Add a third model
  *********************

collect _r_b _r_se,                         ///
        name(MyModels)                      ///
        tag(model[(3)])                     ///
        : logistic heartatk c.age##i.sex i.diabetes

collect AIC=r(S)[1,"AIC"]                   ///
        BIC=r(S)[1,"BIC"],                  ///
        name(MyModels)                      ///
        tag(model[(3)])                     ///
        : estat ic

collect layout (colname#result result[AIC BIC]) (model), name(MyModels)

  ********************************************
  * Use collect style to format the table
  ********************************************

// TURN OFF BASE LEVELS FOR FACTOR VARIABLES
collect style showbase off
// CHANGE THE INTERACTION DELIMITER
collect style row stack, spacer delimiter(" x ")
// REMOVE THE VERTICAL LINE
collect style cell border_block, border(right, pattern(nil))
// FORMAT THE NUMBERS
collect style cell, nformat(%5.2f)
collect style cell result[AIC BIC], nformat(%8.0f)
// PUT PARENTHESES AROUND THE STANDARD ERRORS
collect style cell result[_r_se], sformat("(%s)")
// LABEL AIC AND BIC
collect style header result[AIC BIC], level(label)

collect preview



  *************************
  *use the collect style cell option halign() to center the items and column headers in the table.
  ************************
. collect style cell cell_type[item column-header], halign(center)

. collect preview

  *********************
  * use the collect style header option level() to hide the labels for the row dimension result.
  *********************

collect style header result, level(hide)
collect preview

  ************************************************************************************
  *the collect style column option extraspace to add an extra space between the columns.
  ************************************************************************************
collect style column, extraspace(1)

******************************************
* Export the table to an Adobe PDF document
******************************************

putpdf clear
putpdf begin
putpdf paragraph, font("Calibri Light",26) halign(center)
putpdf text ("Hypertension in the United States")
putpdf paragraph, font("Calibri Light",14) halign(left)
putpdf text ("The National Health and Nutrition Examination Survey (NHANES)")
putpdf paragraph
putpdf text ("Hypertension is a major cause of morbidity and mortality in ")
putpdf text ("the United States.  This report will explore the predictors ")
putpdf text ("of hypertension using the NHANES dataset.")
collect style putpdf, width(60%) indent(1 in)                                ///
        title("Table 3: Logistic Regression Models for Hypertension Status") ///
        note("Note: Odds ratio (standard error)")
putpdf collect
putpdf save MyTable3.pdf, replace

*********************
* save the label, style, and collection
*********************
collect style save "c:\ado\personal\MyLogitStyle", replace
collect label save "c:\ado\personal\MyLogitLabels", replace

*********************************************************
* Apply the saved table label and style to other analyses
*********************************************************

webuse lbw, clear
describe low age smoke ht

table () (command result), command(_r_b _r_se _r_z _r_p _r_ci ///
: logistic low c.age##i.smoke ht)  ///
style(MyLogitStyle, override) label(MyLogitLabels)

**************************
* customize the label of column
**************************

collect label list colname, all



collect label levels colname age "Age", modify
collect label levels colname ht "Hypertension", modify
collect label levels colname smoke "Smoke", modify

collect preview

log close

******************************************************************************************************

log using "D:\Jason\workshop\Stata workship\Stata presentation\stata presentation5_part3.log", replace

******************************
* Aim #3. Shows how to use -graph- commands to generate and customize graphs
*******************************

******************************
* Bar Chart
******************************

use https://www.stata-press.com/data/r17/nlsw88, clear

sum
tab2 wage race
sort race
by race: sum wage, detail

* You can plot mean, median, percentile, sum, count, percent, minimum value, and maximum value

graph bar (mean) wage, over(race)

graph bar (median) wage, over(race)

* The difference between the -by- and -over-option

graph bar wage , by(race)
graph bar wage, over(race)
graph bar wage, over(race) over(collgrad)
graph bar wage, over(race) over(collgrad) over(union)

*******************
* Combine plots
*******************

use https://www.stata-press.com/data/r17/uslifeexp, clear

line le_male year, saving("D:\temp\male.gph", replace)
line le_female year if inrange(year, 1900,1950), saving("D:\temp\female.gph", replace)

* Each graph uses its own x-axis

graph combine "D:\temp\male.gph"  "D:\temp\female.gph", cols(1) saving("D:\temp\combine100.gph", replace)

* Make both graphs use the same x-axis 

graph combine "D:\temp\male.gph"  "D:\temp\female.gph", cols(1) xcommon iscale(.5) saving("D:\temp\combine50.gph", replace)

******************
* Scatter plots
******************



use https://www.stata-press.com/data/r17/sp500, clear

replace volume = volume/1000 

twoway rspike hi low date || ///
line close date || ///
bar volume date, barw(.25) yaxis(2) || ///
in 1/57 ///
, yscale(axis(1) r(900 1400)) ///
yscale(axis(2) r( 9 45)) ///
ytitle(" Price -- High, Low, Close") ///
ytitle(" Volume (millions)", axis(2) bexpand just(left)) ///
legend(off) ///
subtitle("S&P 500", margin(b+2.5)) ///
note("Source: Yahoo!Finance and Commodity Systems, Inc.")

log close


